
前    言

本标准规定了交换用的电子名片的逻辑构成和编码方式，该标准的制定和实施将正确引导电子名片软件的研制与开发。

本标准的附录A是规范性附录。

本标准由国家标准化技术管理委员会提出。

本标准由全国信息技术标准化技术委员会归口。

本标准起草单位:中国电子技术标准化研究所、汉王科技股份有限公司。

本标准主要起草人:
本标准于2009年X月首次发布。
电子名片输出格式通用规范

1　 范围

本标准规定了电子名片输出格式和编码方式。

本标准适用于批量电子名片信息的导出和交换。
2　 规范性引用文件

下列文件中的条款通过本标准的引用而成为本标准的条款。凡是注日期的引用文件，其随后所有的修改单（不包括勘误的内容）或修订版均不适用于本标准，然而，鼓励根据本标准达成协议的各方研究是否可使用这些文件的最新版本。凡是不注日期的引用文件，其最新版本适用于本标准。

GB/T 1988  信息技术 信息交换用七位编码字符集（GB/T 1988-1998, eqv ISO/IEC 646: 1991）

GB/T 15537  信息技术  SGML支持设施  公共文本拥有者标识符的登记规程

GB/T 16264.6  信息技术 开放系统互连 目录 第6部分：选择属性类型

GB 18030  信息技术 信息交换用汉字编码字符集 基本集的扩充

GB/T 19245 电子名片信息交换规范

ISO 8601  数据元素和交换格式 信息交换 日期和时间的表示

RFC 2045  Multipurpose Internet Mail Extensions (MIME) - Part One: Format of Internet Message Bodies
RFC 2047  Multipurpose Internet Mail Extensions (MIME) - Part Three: Message Header Extensions forNon-ASCII Text
RFC 2425  A MIME Content-Type for Directory Information
RFC 2426  vCard MIME Directory Profile
3　 术语和定义

下列术语和定义适用于本标准。

vCard

vCard是IMC制订的电子名片的文件格式标准。它一般附加在电子邮件之后，但也可以用于其它场合（如在因特网上相互交换）。vCard可包含的信息有：姓名、地址信息、电话号码、URL，logo，相片等。

多用途互联网电子邮件扩展 Multipurpose Internet Mail Extensions 简称MIME
最早的HTTP协议中，并没有附加的数据类型信息，所有传送的数据都被客户程序解释为超文本标记语言HTML 文档，而为了支持多媒体数据类型，HTTP协议中就使用了附加在文档之前的MIME数据类型信息来标识数据类型。 

MIME意为多用途Internet邮件扩展，它设计的最初目的是为了在发送电子邮件时附加多媒体数据，让邮件客户程序能根据其类型进行处理。然而当它被HTTP协议支持之后，它的意义就更为显著了。
BINARY数据
在PHOTO，LOGO，SOUND和KEY类型定义中可以出现"binary"值类型数据。"binary"值类型表明该类型值是内联的编码二进制数据。如果出现编码二进制数据，则ENCODING类型参数必须出现在属性定义中，并且编码二进制数据必须使用"B"编码格式，同时单行字符数超过75时应采取[RFC 2425]中5.8.1节所规定的折行方式。这里所说的"B"编码是[RFC 2025]中定义的"BASE64" 编码格式。

8bit数据
所谓8bit数据，指的是满足下列要求的数据：

· 每行字符数不超过998个，行分隔符CRLF不计算在内；

· 不像US-ASCII，十进制数值超过127的字节可以使用；

· NUL不允许出现，CR和LF不允许单独出现，仅能一同出现成为行分隔符CRLF；
4　 电子名片输出格式
4.1　 电子名片文件内容

本标准定义的电子名片输出格式文件内容参考vCard3.0规范，每个vCard对象均以BEGIN:VCARD开始，以END：VCARD结束。一个电子名片文件包含一个或多个vCard对象，文件扩展名为vcf或vcard。

从存储格式、交换格式或解析方式上说，vCard是包含了一个或多个vCard对象的数据流。单个的vCard定义，可以从vCard数据流中识别并进行语法分析。vCard数据流的表现形式可与文件系统、档案管理组成单元形式相一致，也可存在于两个网络终端之间的网络连结，或任何其他的可对字节流解释抽象的数字传输系统。
从概念上讲，vCard写入器指创建vCard数据流的软模块，vCard读入器则为解释vCard数据流的软模块，vCard读入器和写入器，既可集成实现，也可分开应用。至于如何实现这些读写过程，则超出本标准。

4.1.1　 vCard对象所含内容

从逻辑上说，每个vCard对象是单个或多个属性的集合，每个属性具有唯一的属性名。多个属性可以在vCard对象内形成属性组。例如，电话号码和注解可以归为一组，以保持对这个电话的注释。除了属性分组，一个vCard对象还能包括其他的、嵌套vCard对象，这一特点使vCard可以记录和某人或对象相关的其他人或对象。此外，还可以描述由多个vCard对象组成的分发列表或工作组。
4.2　 编码特征

4.2.1　 电子名片对象

电子名片数据流可以包括一至多个电子名片对象。数据流内的单个名片对象由名片分隔符：
[group "."] "BEGIN" ":" "VCARD" 1*CRLF
开始，这个前导字串必须作为数据流或某行中的第一个字符串。
名片对象的结束，必须是：名片对象开始行后面某行，其首位字符段出现

[group "."] "END" ":" "VCARD" 1*CRLF

的特征字符串。
4.2.2　 属性行
每个电子名片表示对象基本组成单位是属性行。每个属性行则由属性名、属性参数列表、属性取值三部分组成，其中属性名与属性参数之间用分号隔开，属性名与属性值之间则用冒号隔开。按照RFC 822所定义的语法规范，属性行可以表现为如下形式：

PropertyName [‘;‘ PropertyParameters] ’:‘ PropertyValue

示例：EMAIL;TYPE=internet:jqpublic@xyz.dom1.com
属性行可以包括一行或多行文本。属性名和属性参数名是预定义好的、大小写非敏感的字符串。 

属性值可以由属性参数列表进一步分类，属性参数列表由一系列的属性参数表达式组成，表达式之间由分号“;”进行分隔，属性参数表达式可以形式为： “name=value”，也可以直接取为参数值字串，但要求取值意义明确，不得有歧义性。
4.2.3　 属性参数

4.2.3.1　 编码

电子名片对象的默认编码是8位, 但可以用属性参数“ENCODING”单独指定。

4.2.3.2　 语言

可以由属性参数 “LANGUAGE”确定。这个属性参数值是一个与RFC 1766规范一致的字串，以作为语言标识。这个属性参数一般可用于任何属性，但对于属性如： PHOTO, LOGO, SOUND, TEL,可能没有意义。汉语的取值可以为zh-CN、 zh-TW、zh-SG或zh-HK。

示例 ：ADR;LANGUAGE=zh-CN:...
示例 ：ADR;LANGUAGE=zh-TW:...
对于其他的属性语句中的LANGUAGE参数取值，指定的中文语言，其作用域仅限于这个参数表达式所在的属性行。

4.2.3.3　 字符集

字符集由字符集属性参数 “CHARSET”取值确定。这个属性参数只能出现在头信息的Content-Type字段中，规定整个名片对象内的属性取值，而不能出现在其他位置。

任何在IANA（Internet Assigned Numbers Authority）上登记的所有汉字的字符集，都可以作为该参数取值。作为中文可能的字符集，主要有：GB12345，GB13000.1，GB18030等。

4.2.4　 属性值类型

属性值类型详见表1：

表1　 属性值类型

	类型
	意义

	Text
	可读文本，其中的逗号和分号前需加反斜线转义。

	Uri
	表示所引用的外部值，其定义见RFC 1738。

	date、time、date-time、utc-offset
	日期、时间，其定义见ISO 8601标准。

	Boolean
	布尔值

	Integer
	以十进制表示的有符号整数，默认为正整数。

	Float
	实数，默认为正实数。


4.3　 vCard 格式

本格式包括两个部分：预定义属性和基本属性。

4.3.1　 预定义属性

4.3.1.1　 开始

表2　 “开始“属性的说明

	标签
	BEGIN

	名称
	开始

	用途
	表示一个内容实体的开始，BEGIN属性的值必须是”VCARD”。

	示例
	BEGIN:VCARD


4.3.1.2　 结束

表3　 “结束“属性的说明

	标签
	END

	名称
	结束

	用途
	表示一个内容实体的结束，END属性的值必须是”VCARD”。

	示例
	END: VCARD


4.3.1.3　 来源名称

表4　 “来源名称“属性的说明

	标签
	NAME

	名称
	来源名称

	用途
	如果在内容中出现NAME属性，那么它的值是一个可以显示的，描述vCard来源的文本。

	编码
	8bit

	取值
	一个text值 

	示例
	NAME:王刚的名片信息


4.3.1.4　 规范

表5　 “规范“属性的说明

	标签
	PROFILE

	名称
	规范

	用途
	用于表示该规范为VCARD规范。

	备注
	如果出现PROFILE属性，那么它的值必须是“VCARD”。

	示例
	PROFILE:vCard


4.3.1.5　 出处

表6　 “出处“属性的说明

	标签
	SOURCE

	名称
	出处

	用途
	如果包含SOURCE类型，它的值提供一些关于如何找到vCard源的信息。

	编码
	8bit

	取值
	一个uri值

	示例
	SOURCE;CONTEXT=LDAP:ldap://ldap.host/cn=Babs%20Jensen,

%20o=Babsco,%20c=US


4.3.2　 基本属性

4.3.2.1　 个人标识属性

4.3.2.1.1　 格式化名称

表7　 “格式化名称“属性的说明

	标签
	FN

	名称
	格式化名称

	用途
	vCard对象的格式化名称，每个vCard对象均需包含该属性。

	编码
	8bit

	取值
	一个text值

	备注
	该属性遵循GB/T 16264.6-2008 公共名属性的语义

	示例
	FN:王刚 


4.3.2.1.2　 名称

表8　 “名称“属性的说明

	标签
	N

	名称
	名称

	用途
	表示这个vCard对象名称的组成部分，每个vCard对象均需包含该属性。

	编码
	8bit

	取值
	一个结构化的text值

	备注
	一个结构化的text值，各个组成部分可以用分号分隔，每个组成部分内部可以用逗号分隔。这个属性值的第一个字段为姓氏，第二个字段为人名，第三个字段为本人的英文名称或拼音注释，第四个字段为学位,第五个字段可以表示为职称、荣誉头衔、尊称等多种名称要素。其中第三、四、五字段可有可无，亦可以只有一个字段,但分隔符（即分号";"）不能少。

	示例
	N:王;刚;;;

N:Public;John;Quinlan;Mr.;Esq.


4.3.2.1.3　 别名

表9　 “别名“属性的说明

	标签
	NICKNAME

	名称
	别名

	用途
	指定vCard对象的别名。

	编码
	8bit

	取值
	一个或多个用逗号分隔的text值

	备注
	

	示例
	NICKNAME:小马


4.3.2.1.4　 图像

表10　 “图像“属性的说明

	标签
	PHOTO

	名称
	图像

	用途
	vCard对象的图像信息

	编码
	如果使用二进制数据表示图片信息，那么需使用ENCODING=b；如果是引用外部的URI值，则编码为8 bit，此时无需ENCODING参数。

	取值
	默认为二进制图像数据，也可为指示外部图像数据的uri值。

	备注
	通过属性参数TYPE指定图像格式。它的值应为IANA注册的格式，或者是一种非标准的格式。

	示例
	PHOTO;ENCODING=b;TYPE=JPEG:MIICajCCAdOgAwIBAgICBEUwDQYJKoZIhvcN
AQEEBQAwdzELMAkGA1UEBhMCVVMxLDAqBgNVBAoTI05ldHNjYXBlIENvbW11bm
ljYXRpb25zIENvcnBvcmF0aW9uMRwwGgYDVQQLExNJbmZvcm1hdGlvbiBTeXN0

PHOTO;VALUE=uri:http://www.abc.com/pub/photos/jqpublic.gif


4.3.2.1.5　 生日

表11　 “生日“属性的说明

	标签
	BDAY

	名称
	生日

	用途
	指定了名片个体的出生日期

	编码
	8bit

	取值
	默认为一个date值，也可为一个date-time值。

	备注
	

	示例
	BDAY:1996-06-15


4.3.2.2　 邮政地址属性

4.3.2.2.1　 邮递地址

表12　 “邮递地址“属性的说明

	标签
	ADR

	名称
	邮递地址

	用途
	是一个组合，用来表示vCard对象的邮递地址信息

	编码
	8bit

	取值
	用分号分开的结构化的text值 

	备注
	该结构化的text值由一系列用分号分隔的地址子部分组成。这些子部分必须按着如下的顺序排列：邮政信箱，扩充地址，街道地址，当地地址，所处地区，邮政编码，国家名称。当某些子部分没有时，相应的分隔符仍需保留。其中，一个子部分可以包括多个值，用逗号分隔。

可以通过属性参数TYPE指定邮递地址的类型。这个属性参数可以选取以下的值:

dom：国内地址

intl：国际地址（默认）

postal：邮政投递地址（默认）

parcel：包裹投递地址（默认）

home：家庭投递地址

work：工作投递地址（默认）

pref：首选的地址（当有多个地址时）

	示例
	ADR; TYPE=dom,home,postal,parcel:;;街道地址;深圳;广东;433330;中国


4.3.2.2.2　 邮寄标签

表13　 “邮寄标签“属性的说明

	标签
	LABEL

	名称
	邮寄标签

	用途
	用于指定与邮递地址对应的格式化文本

	编码
	8bit

	取值
	一个text值

	备注
	可以通过属性参数TYPE指定邮递地址的类型。这个属性参数可以选取以下的值:

dom：国内地址

intl：国际地址（默认）

postal：邮政投递地址（默认）

parcel：包裹投递地址（默认）

home：家庭投递地址

work：工作投递地址（默认）

pref：首选的地址

和ADR不同的是 ADR的值是用分号分隔的数据，而LABEL就是一个格式化的文本。

	示例
	LABEL;TYPE=dom,home,postal,parcel:海淀北大街123号\,海淀区\,北京\,100080


4.3.2.3　 通信地址属性

4.3.2.3.1　 电话

表14　 “电话“属性的说明

	标签
	TEL

	名称
	电话

	用途
	指定一个电话号码

	编码
	8bit

	取值
	一个phone-number值，原则上应依据GB/T 16264.6 电话号码属性规定

	备注
	可以通过属性参数TYPE指定电话的类型。这个属性参数可以选取以下的值:
    home：家庭电话

    msg：支持语音留言功能的电话

    work：工作电话

    pref：首选的电话

voice：语音电话（默认）

fax: 传真号码

    cell: 蜂窝式移动电话

    video: 视频电话

    pager: 寻呼机号码

bbs: 公告板服务号码

modem: 连接了调制解调器的电话号码

    car: 车载电话

    isdn: 开通了ISDN服务的电话号码

    pcs: 个人通信服务电话
    assistant: 助理电话
    telegraph: 电报电话
    tty/tdd: tty/tdd电话

	示例
	TEL;TYPE=work,voice,pref:++86-132-666-1234


4.3.2.3.2　 电子邮件

表15　 “电子邮件“属性的说明

	标签
	EMAIL

	名称
	电子邮件

	用途
	表示一个电子邮件地址

	编码
	8bit

	取值
	一个text值

	备注
	可以通过属性参数TYPE指定电子邮件地址的类型。这个属性参数可以选取以下的值:

internet：internet 类型地址（默认）

x400：X.400 类型地址

pref：首选的电子邮件地址（当有多个地址时）

tlx： 表示电报电话

	示例
	EMAIL;TYPE=internet:jqpublic@xyz.dom1.com


4.3.2.3.3　 电子邮件软件

表16　 “电子邮件软件“属性的说明

	标签
	MAILER

	名称
	电子邮件软件

	用途
	指明vCard对象中的个人所使用的电子邮件软件类型

	编码
	8bit

	取值
	一个text值

	备注
	

	示例
	MAILER:PigeonMail 2.1


4.3.2.4　 地理属性

4.3.2.4.1　 时区

表17　 “时区“属性的说明

	标签
	TZ

	名称
	时区

	用途
	表示时区信息

	编码
	8bit

	取值
	缺省为一个utc-offset值，也可以是一个text值

	备注
	

	示例
	TZ:-05:00

TZ;VALUE=text:-05:00\;北京时间


4.3.2.4.2　 经纬度

表18　 “经纬度“属性的说明

	标签
	GEO

	名称
	经纬度

	用途
	表示地理位置信息

	编码
	8bit

	取值
	一个由两个用分号分隔的float值所构成的结构化的值 

	备注
	纬度在前，经度在后。

	示例
	GEO:37.386013;-122.082932 


4.3.2.5　 组织机构或单位属性

4.3.2.5.1　 头衔

表19　 “头衔“属性的说明

	标签
	TITLE

	名称
	头衔

	用途
	指定与名片对象相关的个人在组织内的工作头衔，职位或作用。

	编码
	8bit

	取值
	一个text值

	备注
	基于 GB/T 16264.6标准中有关 TITLE属性定义

	示例
	TITLE:研发经理


4.3.2.5.2　 业务类别

表20　 “业务类别“属性的说明

	标签
	ROLE

	名称
	业务类别

	用途
	指定组织系统内电子名片对象所属的角色、职业或业务类别。

	编码
	8bit

	取值
	一个text值

	备注
	基于GB/T 16264标准中的业务类别解释属性

	示例
	ROLE:行政管理


4.3.2.5.3　 徽标

表21　 “徽标“属性的说明

	标签
	LOGO

	名称
	徽标

	用途
	表示公司logo，是一个图像信息

	编码
	binary 或 8bit, 如果使用二进制数据表示图片信息，那么需使用ENCODING=b。

	取值
	默认为二进制图像数据，也可为指示外部图像数据的uri值。

	备注
	通过属性参数TYPE指定图像格式。它的值应为IANA注册的格式，或者是一种非标准的格式。

	示例
	LOGO;VALUE=uri:http://www.abc.com/pub/logos/abccorp.jpg

LOGO;ENCODING=b;TYPE=JPEG:MIICajCCAdOgAwIBAgICBEUwDQYJKoZIhvcN         AQEEBQAwdzELMAkGA1UEBhMCVVMxLDAqBgNVBAoTI05ldHNjYXBlIENvbW11bm

ljYXRpb25zIENvcnBvcmF0aW9uMRwwGgYDVQQLExNJbmZvcm1hdGlvbiBTeXN0


4.3.2.5.4　 代理

表22　 “代理“属性的说明

	标签
	AGENT

	名称
	代理

	用途
	指定代表与vCard对象相关的个人或资源的另一个个体的信息

	编码
	8bit

	取值
	默认为一个vCard值，也可为一个text值或uri值。

	备注
	这个属性代表着一个公司的地区性负责人、助理、或秘书职位。代理属性的主要特征是它表示一个具有独立地址的个人或事物。

	示例
	AGENT:BEGIN:VCARD\nFN:Joe Friday\nTEL:+1-919-555-7878\n

 TITLE:Area Administrator\, Assistant\n EMAIL\;TYPE=INTERN\n

 ET:jfriday@host.com\nEND:VCARD\n


4.3.2.5.5　 组织

表23　 “组织“属性的说明

	标签
	ORG

	名称
	组织

	用途
	表示一个组织机构或单位的名称

	编码
	8bit

	取值
	一个由分号分隔的子部分组成的结构化text值

	备注
	基于GB/T 16264.6 组织名属性及组织下级单位属性。这个属性值包括的各级组织成分，用分号分隔。各级组织成分，主要指组织名称（第一个字段）、组织下级单位（第二个字段）、更下级的单位名称（第三个字段），依此递推。

	示例
	ORG:汉王科技;研发中心;OCR软件部


4.3.2.6　 解释性信息属性

4.3.2.6.1　 类别

表24　 “类别“属性的说明

	标签
	CATEGORIES

	名称
	类别

	用途
	vCard的应用分类信息

	编码
	8bit

	取值
	一个或多个用逗号分隔的text值

	备注
	

	示例
	CATEGORIES:因特网,信息技术


4.3.2.6.2　 注释

表25　 “注释“属性的说明

	标签
	NOTE

	名称
	注释

	用途
	vCard对象相关的补充信息或注释

	编码
	8bit

	取值
	一个text值

	备注
	基于GB/T 16264.6描述属性定义

	示例
	NOTE:注释信息.


4.3.2.6.3　 产品ID

表26　 “产品ID“属性的说明

	标签
	PRODID

	名称
	产品ID

	用途
	指定创建vCard对象的产品的ID

	编码
	8bit

	取值
	一个text值

	备注
	生成方法参见GB/T 15537

	示例
	PRODID: 12312784-974320-819738238730-8634


4.3.2.6.4　 最近修正时间

表27　 “最近修正时间“属性的说明

	标签
	REV

	名称
	最近修正时间

	用途
	表示出当前vCard的最近修正时间

	编码
	8bit

	取值
	默认为一个date-time值，也可为一个date值。

	备注
	

	示例
	REV:1995-10-31T22:27:10Z

REV:1997-11-15


4.3.2.6.5　 排序名称

表28　 “排序名称“属性的说明

	标签
	SORT-STRING

	名称
	排序名称

	用途
	指定姓氏或名称以用于对FN和N属性排序

	编码
	8bit

	取值
	一个text值

	备注
	

	示例
	FN:Rene van der Harten

N:van der Harten;Rene;J.;Sir;R.D.O.N.

SORT-STRING:Harten


4.3.2.6.6　 声音

表29　 “声音“属性的说明

	标签
	SOUND

	名称
	声音

	用途
	指定vCard的数字声音信息，缺省是指vCard对象的name属性的发音信息

	编码
	binary 或 8bit, 如果使用二进制数据表示声音信息，那么需使用ENCODING=b。

	取值
	默认为二进制声音数据，也可为指示外部声音数据的uri值。

	备注
	可以通过属性参数TYPE指定采用的声音的格式。它的值应为IANA注册的格式，或者是一种非标准的格式。

	示例
	SOUND;TYPE=BASIC;VALUE=uri:CID:JOHNQPUBLIC.part8.19960229T0800
00.xyzMail@host1.com

SOUND;TYPE=BASIC;ENCODING=b:MIICajCCAdOgAwIBAgICBEUwDQYJKoZIhvcN         AQEEBQAwdzELMAkGA1UEBhMCVVMxLDAqBgNVBAoTI05ldHNjYXBlIENvbW11bm         ljYXRpb25zIENvcnBvcmF0aW9uMRwwGgYDVQQLExNJbmZvcm1hdGlvbiBTeXN0


4.3.2.6.7　 唯一标识符

表30　 “唯一标识符“属性的说明

	标签
	UID

	名称
	唯一标识符

	用途
	指定一个标识该vCard相关的个人或资源的全球唯一标识符

	编码
	8bit

	取值
	一个text值

	备注
	生成方法参见GB/T 15537

	示例
	UID: 49878357-347530-347093847304-8347


4.3.2.6.8　 统一资源定位器

表31　 “统一资源定位器“属性的说明

	标签
	URL

	名称
	统一资源定位器

	用途
	指定该vCard相关的网址

	编码
	8bit

	取值
	一个uri值

	备注
	

	示例
	URL:http://www.hanwang.com.cn


4.3.2.6.9　 版本

表32　 “版本“属性的说明

	标签
	VERSION

	名称
	版本

	用途
	指定该vCard所使用的vCard规范的版本

	编码
	8bit

	取值
	一个text值

	备注
	

	示例
	VERSION:3.0


4.3.2.7　 名片安全属性
4.3.2.7.1　 访问级别

表33　 “访问级别“属性的说明

	标签
	CLASS

	名称
	访问级别

	用途
	指定vCard对象的访问级别

	编码
	8bit

	取值
	一个text值

	备注
	访问级别需要参考目录服务的访问级别

	示例
	CLASS:PUBLIC

CLASS:PRIVATE

CLASS:CONFIDENTIAL


4.3.2.7.2　 公用密钥

表34　 “公用密钥“属性的说明

	标签
	KEY

	名称
	公用密钥

	用途
	指定vCard的公共钥匙值（加密解密时使用）或者是授权认证

	编码
	binary 或 8bit

	取值
	默认为二进制数据，也可为text值。

	备注
	

	示例
	KEY;ENCODING=b:MIICajCCAdOgAwIBAgICBEUwDQYJKoZIhvcNAQEEBQA
wdzELMAkGA1UEBhMCVVMxLDAqBgNVBAoTI05ldHNjYXBlIENbW11bmljYX       Rpb25zIENvcnBvcmF0aW9uMRwwGgYDVQQLExNJbmZvcm1hdGlvbiBTeXN0         ZW1zMRwwGgYDVQQDExNyb290Y2EubmV0c2NhcGUuY29tMB4XDTk3MDYwNj         E5NDc1OVoXDTk3MTIwMzE5NDc1OVowgYkxCzAJBgNVBAYTAlVTMSYwJAYD         VQQKEx1OZXRzY2FwZSBDb21tdW5pY2F0aW9ucyBDb3JwLjEYMBYGA1UEAx         MPVGltb3RoeSBBIEhvd2VzMSEwHwYJKoZIhvcNAQkBFhJob3dlc0BuZXRz         Y2FwZS5jb20xFTATBgoJkiaJk/IsZAEBEwVob3dlczBcMA0GCSqGSIb3DQ         EBAQUAA0sAMEgCQQC0JZf6wkg8pLMXHHCUvMfL5H6zjSk4vTTXZpYyrdN2         dXcoX49LKiOmgeJSzoiFKHtLOIboyludF90CgqcxtwKnAgMBAAGjNjA0MB         EGCWCGSAGG+EIBAQQEAwIAoDAfBgNVHSMEGDAWgBT84FToB/GV3jr3mcau         +hUMbsQukjANBgkqhkiG9w0BAQQFAAOBgQBexv7o7mi3PLXadkmNP9LcIP         mx93HGp0Kgyx1jIVMyNgsemeAwBM+MSlhMfcpbTrONwNjZYW8vJDSoi//y         rZlVt9bJbs7MNYZVsyF1unsqaln4/vy6Uawfg8VUMk1U7jt8LYpo4YULU7         UZHPYVUaSgVttImOHZIKi4hlPXBOhcUQ==



（规范性附录）
电子名片输出格式的语法规则

A.1　 语法规则

下列给出电子名片输出格式通用规范的语法规则，用于帮助开发者建造电子名片的分析器Parser。这个语法引自RFC 2426第4节所述内容，但有修改。

   ;*******************************************

   ; Commonly Used Literal Definition

   ;*******************************************

   ALPHA        = %x41-5A / %x61-7A

        ; Latin Capital Letter A-Latin Capital Letter Z /

        ; Latin Small Letter a-Latin Small Letter z

   CHAR         = %x01-7F

        ; Any C0 Controls and Basic Latin, excluding NULL from

        ; Code Charts, pages 7-6 through 7-9 in [UNICODE]

   CR           = %x0D

        ; Carriage Return

   LF           = %0A

        ; Line Feed

   CRLF         = CR LF

        ; Internet standard newline

   ;CTL         = %x00-1F / %x7F

        ; Controls. Not used, but referenced in comments.

   DIGIT        = %x30-39

        ; Digit Zero-Digit Nine

   DQUOTE       = %x22

        ; Quotation Mark

   HTAB         = %x09

        ; Horizontal Tabulation

   SP           = %x20

        ; space

   VCHAR        = %x21-7E

        ; Visible (printing) characters

   WSP          = SP / HTAB

        ; White Space

   ;*******************************************

   ; Basic vCard Definition

   ;*******************************************

   vcard_entity = 1*(vcard)

   vcard        = [group "."] "BEGIN" ":" "VCARD" 1*CRLF

                  1*(contentline)

        ;A vCard object MUST include the VERSION, FN and N types.

                  [group "."] "END" ":" "VCARD" 1*CRLF

   contentline  = [group "."] name *(";" param ) ":" value CRLF

        ; When parsing a content line, folded lines must first

        ; be unfolded according to the unfolding procedure

        ; described above. When generating a content line, lines

        ; longer than 75 characters SHOULD be folded according to

        ; the folding procedure described in [MIME DIR].

   group        = 1*(ALPHA / DIGIT / "-")

   name         = iana-token / x-name

        ; Parsing of the param and value is

        ; based on the "name" or type identifier

        ; as defined in ABNF sections below

   iana-token   = 1*(ALPHA / DIGIT / "-")

        ; vCard type or parameter identifier registered with IANA

   x-name       = "X-" 1*(ALPHA / DIGIT / "-")

        ; Reserved for non-standard use

   param        = param-name "=" param-value *("," param-value)

   param-name   = iana-token / x-name

   param-value  = ptext / quoted-string

   ptext        = *SAFE-CHAR

   value        = *VALUE-CHAR

   quoted-string = DQUOTE QSAFE-CHAR DQUOTE

   NON-ASCII    = %x80-FF

        ; Use is restricted by CHARSET parameter

        ; on outer MIME object (UTF-8 preferred)

   QSAFE-CHAR   = WSP / %x21 / %x23-7E / NON-ASCII

        ; Any character except CTLs, DQUOTE

   SAFE-CHAR    = WSP / %x21 / %x23-2B / %x2D-39 / %x3C-7E / NON-ASCII

        ; Any character except CTLs, DQUOTE, ";", ":", ","

   VALUE-CHAR   = WSP / VCHAR / NON-ASCII

        ; Any textual character

   ;*******************************************

   ; vCard Type Definition

   ;

   ; Provides type-specific definitions for how the

   ; "value" and "param" are defined.

   ;*******************************************

   ;For name="NAME"

   param        = ""

        ; No parameters allowed

   value        = text-value

   ;For name="PROFILE"

   param        = ""

        ; No parameters allowed

   value        = text-value

        ; Value MUST be the case insensitive value "VCARD

   ;For name="SOURCE"

   param        = source-param

        ; Only source parameters allowed

   value        = uri

   source-param = ("VALUE" "=" "uri")

                / ("CONTEXT" "=" "word")

        ; Parameter value specifies the protocol context

        ; for the uri value.

                / (x-name "=" *SAFE-CHAR)

   ;For name="FN"

   ;This type MUST be included in a vCard object.

   param        = text-param

        ; Text parameters allowed

   value        = text-value

   ;For name="N"

   ;This type MUST be included in a vCard object.

   param        = text-param

        ; Text parameters allowed

   value        = n-value

   n-value      = 0*4(text-value *("," text-value) ";")

                  text-value *("," text-value)

        ; Family; Given; Middle; Prefix; Suffix.

        ; Example: Public;John;Quincy,Adams;Reverend Dr. III

   ;For name="NICKNAME"

   param        = text-param

        ; Text parameters allowed

   value        = text-value-list

   ;For name="PHOTO"

   param        = img-inline-param

        ; Only image parameters allowed

   param        =/ img-refer-param

        ; Only image parameters allowed

   value        = img-inline-value

        ; Value and parameter MUST match

   value        =/ img-refer-value

        ; Value and parameter MUST match

   ;For name="BDAY"

   param        = ("VALUE" "=" "date")

        ; Only value parameter allowed

   param        =/ ("VALUE" "=" "date-time")

        ; Only value parameter allowed

   value        = date-value

        ; Value MUST match value type

   value        =/ date-time-value

        ; Value MUST match value type

   ;For name="ADR"

   param        = adr-param / text-param

        ; Only adr and text parameters allowed

   value        = adr-value

   ;For name="LABEL"

   param        = adr-param / text-param

        ; Only adr and text parameters allowed

   value        = text-value

   ;For name="TEL"

   param        = tel-param

        ; Only tel parameters allowed

   value        = phone-number-value

   tel-param    = "TYPE" "=" tel-type *("," tel-type)

   tel-type     = "HOME" / "WORK" / "PREF" / "VOICE" / "FAX" / "MSG"

                / "CELL" / "PAGER" / "BBS" / "MODEM" / "CAR" / "ISDN"

                / "VIDEO" / "PCS" / iana-token / x-name

        ; Values are case insensitive

   ;For name="EMAIL"

   param        = email-param

        ; Only email parameters allowed

   value        = text-value

   email-param  = "TYPE" "=" email-type ["," "PREF"]

        ; Value is case insensitive

   email-type   = "INTERNET" / "X400" / iana-token / x-name

        ; Values are case insensitive

   ;For name="MAILER"

   param        = text-param

        ; Only text parameters allowed

   value        = text-value

   ;For name="TZ"

   param        = ""

        ; No parameters allowed

   value        = utc-offset-value

   ;For name="GEO"

   param        = ""

        ; No parameters allowed

   value        = float-value ";" float-value

   ;For name="TITLE"

   param        = text-param

        ; Only text parameters allowed

   value        = text-value

   ;For name="ROLE"

   param        = text-param

        ; Only text parameters allowed

   value        = text-value

   ;For name="LOGO"

   param        = img-inline-param / img-refer-param

        ; Only image parameters allowed

   value        = img-inline-value / img-refer-value

        ; Value and parameter MUST match

   ;For name="AGENT"

   param        = agent-inline-param

   param        =/ agent-refer-param

   value        = agent-inline-value

        ; Value and parameter MUST match

   value        =/ agent-refer-value

        ; Value and parameter MUST match

   agent-inline-param = ""

        ; No parameters allowed

   agent-refer-param = "VALUE" "=" "uri"

        ; Only value parameter allowed

   agent-inline-value = text-value

        ; Value MUST be a valid vCard object

   agent-refer-value = uri

        ; URI MUST refer a valid vCard object

   ;For name="ORG"

   param        = text-param

        ; Only text parameters allowed

   value        = org-value

   org-value    = *(text-value ";") text-value

        ; First is Organization Name, remainder are Organization Units.

   ;For name="CATEGORIES"

   param        = text-param

        ; Only text parameters allowed

   value        = text-value-list

   ;For name="NOTE"

   param        = text-param

        ; Only text parameters allowed

   value        = text-value

   ;For name="PRODID"

   param        = ""

        ; No parameters allowed

   value        = text-value

   ;For name="REV"

   param        = ["VALUE" =" "date-time"]

        ; Only value parameters allowed. Values are case insensitive.

   param        =/ "VALUE" =" "date"

        ; Only value parameters allowed. Values are case insensitive.

   value        = date-time-value

   value        =/ date-value

   ;For name="SORT-STRING"

   param        = text-param

        ; Only text parameters allowed

   value        = text-value

   ;For name="SOUND"

   param        = snd-inline-param

        ; Only sound parameters allowed

   param        =/ snd-refer-param

        ; Only sound parameters allowed

   value        = snd-inline-value

        ; Value MUST match value type

   value        =/ snd-refer-value

        ; Value MUST match value type

   snd-inline-value     = binary-value

        ; Value MUST be "b" encoded audio content

   snd-inline-param     = ("VALUE" "=" "binary")

                        / ("ENCODING" "=" "b")

                        / ("TYPE" "=" *SAFE-CHAR)

        ; Value MUST be an IANA registered audio type

   snd-refer-value      = uri

        ; URI MUST refer to audio content of given type

   snd-refer-param      = ("VALUE" "=" "uri")

                        / ("TYPE" "=" word)

        ; Value MUST be an IANA registered audio type

   ;For name="UID"

   param        = ""

   param        = "TYPE" "=" (iana-token / x-name)

        ; No parameters allowed

   value        = text-value

   ;For name="URL"

   param        = ""

        ; No parameters allowed

   value        = uri

   ;For name="VERSION"

   ;This type MUST be included in a vCard object.

   param        = ""

        ; No parameters allowed

   value        = text-value

        ; Value MUST be "3.0"

   ;For name="CLASS"

   param        = ""

        ; No parameters allowed

   value        = "PUBLIC" / "PRIVATE" / "CONFIDENTIAL"

                / iana-token / x-name

        ; Value are case insensitive

   ;For name="KEY"

   param        = key-txt-param

        ; Only value and type parameters allowed

   param        =/ key-bin-param

        ; Only value and type parameters allowed

   value        = text-value

   value        =/ binary-value

   key-txt-param = "TYPE" "=" keytype

   key-bin-param = ("VALUE" "=" "binary")

                 / ("TYPE" "=" keytype)

                 / ("ENCODING" "=" "b")

        ; Value MUST be a "b" encoded key or certificate

  keytype      = "X509" / "PGP" / iana-token / x-name

        ; Values are case insensitive

   ;For name="X-" non-standard type

   param        = text-param / (x-name "=" param-value)

        ; Only text or non-standard parameters allowed

   value        = text-value

   ;*******************************************

   ; vCard Commonly Used Parameter Definition

   ;*******************************************

   text-param   = ("VALUE" "=" "ptext")

                / ("LANGUAGE" "=" langval)

                / (x-name "=" param-value)

   langval      = <a language string as defined in RFC 1766>

   img-inline-value     = binary-value

        ;Value MUST be "b" encoded image content

   img-inline-param     = ("VALUE" "=" "binary")

                        / ("ENCODING" "=" "b")

                        / ("TYPE" "=" param-value)

        ;TYPE value MUST be an IANA registered image type

   img-refer-value = uri

        ;URI MUST refer to image content of given type

   img-refer-param      = ("VALUE" "=" "uri")

                        / ("TYPE" "=" param-value)

        ;TYPE value MUST be an IANA registered image type

   adr-param    = ("TYPE" "=" adr-type *("," adr-type))

                / (text-param)

   adr-type     = "dom" / "intl" / "postal" / "parcel" / "home"

                / "work" / "pref" / iana-token / x-name

   adr-value    = 0*6(text-value ";") text-value

        ; PO Box, Extended Address, Street, Locality, Region, Postal

        ; Code, Country Name

   ;*******************************************

   ; vCard Type Value Definition

   ;*******************************************

   text-value-list      = 1*text-value *("," 1*text-value)

   text-value   = *(SAFE-CHAR / ":" / DQUOTE / ESCAPED-CHAR)

   ESCAPED-CHAR = "\\" / "\;" / "\," / "\n" / "\N")

        ; \\ encodes \, \n or \N encodes newline

        ; \; encodes ;, \, encodes ,

   binary-value = <A "b" encoded text value as defined in [RFC 2047]>

   date-value   = <A single date value as defined in [MIME-DIR]>

   time-value   = <A single time value as defined in [MIME-DIR]>

   date-time-value = <A single date-time value as defined in [MIME-DIR]

   float-value  = <A single float value as defined in [MIME-DIR]>

   phone-number-value = <A single text  value as defined in [CCITT

                         E.163] and [CCITT X.121]>

   uri-value    = <A uri value as defined in [MIME-DIR]>

   utc-offset-value = ("+" / "-") time-hour ":" time-minute

   time-hour    = 2DIGIT                ;00-23

time-minute  = 2DIGIT                ;00-59

